

Megoldás Az 1. ábrán bemutatott program könnyen érthető, részletes, hibátlan megoldásnak tűnik. Ellenőrizzük működését az alábbi adatokkal is:

AX	AY	AR	BX	BY	BR
-4	5	2	-7	1	3
7	16	2	2	4	5

```
5 CLS: PRINT "KOROK KOLCSONOS HELYZETENEK MEGHATAROZASA"
10 PRINT "ADD MEG AZ ELSO KOR KOZEPPONTJANAK KOORDINATAIT
    ES SUGARAT!"
20 INPUT AX, AY, AR
30 PRINT "ADD MEG A MASODIK KOR ADATAIT IS!"
40 INPUT BX, BY, BR
45 IF AR<0 OR BR<0 THEN10
50 T=SQR((AX-BX) * (AX-BX)+(AY-BY) * (AY-BY))
60 IF T > 0 THEN 90
70 IF AR=BR THEN PRINT "A KET KOR AZONOS!": STOP
80 PRINT "A KET KOR KONCENTRIKUS!": STOP
90 IF T<ABS(AR-BR) THEN PRINT "A NAGYOBBIK KOR TARTALMAZZA
    A KISEBBIKET": STOP
100 IF T=ABS(AR-BR) THEN PRINT "A KISEBBIK KOR BELULROL
    ERINTI A NAGYOBBIKAT": STOP
110 IF T=AR+BR THEN PRINT "A KET KOR KIVULROL ERINTI
    EGYMAST": STOP
120 IF T>AR+BR THEN PRINT "A KET KOR ELKERULI EGYMAST": STOP
130 IF "A KET KOR METSZI EGYMAST,
140 IF T=AR OR T=BR THEN PRINT "AZ EGYIK KOR KOZEPPONTJA
    RAJTA VAN A MASIK KORON": STOP
150 IF T<AR AND T<BR THEN PRINT "MINDKET KOR TARTALMAZZA
    A MASIK KOZEPPONTJAT": STOP
160 IF T<AR OR T<BR THEN PRINT "A NAGYOBBIK KOR TARTALMAZZA
    A KISEBBIK KOZEPPONTJAT": STOP
170 PRINT "EGYIK SEM TARTALMAZZA A MASIK KOZEPPONTJAT": STOP
```

1. ábra

Ezekben az esetekben az érintkező köröket programunk közös pont nélküli, ill. metsző köröknek nyilvánítja. Mi ennek a magyarázata? A négyzetgyökvonást még négyzetszámok esetén sem végzi el teljes pontossággal a gép – így némi eltérést észlel T és $AR + BR$, ill. T és $ABS(AR - BR)$ között. Ezért ír téves választ. Ha úgy módosítjuk programunkat, hogy gyökvonás helyett szorzással elvégzett négyzetre emelést írunk elő, egész adatok esetén korrekt

eredményt kapunk. A 2. ábra csak a megváltozó, ill. új program-sorokat tartalmazza.

```
50 T=(AX-BX) * (AX-BX) +(AY-BY) * (AY-BY)
70 IF AR=BR THEN PRINT " A KET KOR AZONOS! ": GOTO 200
80 PRINT " A KET KOR KONCENTRIKUS!": GOTO 200
90 D=(AR-BR) * (AR-BR): IF T<D THEN PRINT "A NAGYOBBIK KOR
    TARTALMAZZA A KISEBBIKET": GOTO 200
100 IF T=D THEN PRINT "A KISEBBIK KOR BELULROL ERINTI A
    NAGYOBBIKAT": GOTO 200
110 E=(AR+BR) * (AR+BR): IF T=E THEN PRINT "A KET KOR KIVULROL
    ERINTI EGYMAST": GOTO 200
120 IF T>E THEN PRINT "A KET KOR ELKERULI EGYMAST": GOTO 200
140 IF T=AR*AR OR T=BR*BR THEN PRINT "AZ EGYIK KOR KOZEPPONTJA
    RAJTA VAN A MASIK KORON ": GOTO 200
150 IF T<AR * AR and T<BR*BR THEN PRINT "MINDKET KOR TARTALMAZZA
    A MASIK KOZEPPONTJAT": GOTO 200
160 IF T<AR*AR OR T<BR*BR THEN PRINT A NAGYOBBIK KOR
    TARTALMAZZA A KISEBBIK KOZEPPONTJAT": GOTO 200
170 PRINT " EGYIK SEM TARTALMAZZA A MASIK KOZEPPONTJAT"
200 A$ = INKEY$
210 IF INKEY$ = ""THEN 210 ELSE 5
```

2. ábra

Megjegyzés. STOP utasítás után CONT parancsra folytatódik a program futása. Az első programváltozat esetén ilyenkor oda nem illő válaszokat ad a gép. Ennek elkerülésére írtuk be a GOTO 200 utasításokat. A RUN parancs ismételt begépelését egy tetszőleges billentyű benyomása helyettesíti, ha beírjuk a KSz. 10. megoldásából már ismert 200 és 210 jelű sorokat. A 200-as sor az esetleges előre lenyomott billentyűt hatástalanítja, hogy a képernyőre írt válasz elolvasására legyen időnk.